Esercizio 2: Cerchi e triangoli

Il problema richiede di definire la posizione dei K cerchi, la posizione di un triangolo che li contenga e la posizione di un cerchio che contenga il triangolo. Servono quindi le seguenti variabili:

- due coordinate cartesiane, che indichiamo con *xc* e *yc* per ciascuno dei K cerchi piccoli;
- tre parametri, che indichiamo con *a*, *b* e *c* per ciascuna delle tre rette che contengono i lati del triangolo;
- tre coppie di coordinate cartesiane, che indichiamo con *xv* e *yv*, per i vertici del triangolo;
- una coppia di coordinate cartesiane, che indichiamo con *xx* e *yy*, per il centro del cerchio grande;
- il raggio del cerchio grande, che indichiamo con *r*.

L'obiettivo è semplicemente quello di minimizzare *r*, che equivale a minimizzare il suo quadrato (dato che nel vincolo del contenimento del triangolo nel cerchio compare r^2.

Senza perdita di generalità il cerchio grande può essere fissato nell'origine e questo fa scomparire le variabili *xx* e *yy*.

E' importante ricordare che tutte le variabili del problema sono da dichiarare come variabili libere.

I vincoli di non sovrapposizione tra i cerchi piccoli si esprimono imponendo che la distanza tra centri di cerchi diversi sia maggiore o uguale a 2 (dato che il loro raggio è pari a 1) o equivalentemente che il quadrato della distanza deve essere maggiore o uguale a 4. Quindi:

$$\left(xc(i)-xc(j)\right)^2+\left(xc(i)-xc(j)\right)^2 \geq 2^2 \;\forall\, i,j : i<j$$

che nella sintassi di Lingo equivale a

```
@for(cerchi(i):
     @for(cerchi(j) | (i#LT#j):
          (xc(i)-xc(j))^2+(yc(i)-yc(j))^2 >= 4
        )
    );
```

Grazie alla condizione *(i#LT#j)* Il vincolo viene imposto solo una volta per ogni coppia di cerchi diversi. Usare la condizione *(i#NE#j)* non sarebbe sbagliato, ma inserirebbe lo stesso vincolo inutilmente due volte per ogni coppia. Non inserire la condizione sarebbe sbagliato perché la distanza di un punto da sé stesso non può mai essere maggiore o uguale a 4 e quindi i vincoli non sarebbero mai soddisfacibili.

Per definire il triangolo ed imporre che sia contenuto nel cerchio grande basterebbero le coordinate dei suoi tre vertici, ma per imporre che esso contenga i cerchi piccoli, serve anche l'equazione delle rette dei lati. Affinché le terne di parametri *(a,b,c)* rappresentino davvero delle rette nel piano, occorre normalizzare i coefficienti *a* e *b* con il vincolo:

$$\sqrt{a(v)^2+b(v)^2}=1 \;\forall\, v=1..3$$

che nella sintassi di Lingo equivale a

```
@for(rette(v): a(v)^2 + b(v)^2 = 1);
```

Le coordinate dei vertici del triangolo si trovano all'intersezione delle coppie di rette. Quindi, assumendo che ogni valore dell'indice *v=1..3* venga usato per rappresentare un lato ed il vertice opposto, si hanno le seguenti condizioni di passaggio di una retta di parametri *(a(v),b(v),c(v))* per un punto di coordinate *(xv(u),yv(u))*:

$$a(v)xv(u)+b(v)yv(u)+c(v)=0\,\forall u,v:u\neq v$$

che valgono per ciascun lato e per ciascun vertice che *non* sia quello opposto. Quindi *u* e *v* sono entrambi indici da 1 a 3 ma devono essere diversi tra loro. Nella sintassi di Lingo:

```
@for(rette(v):
     @for(rette(u) | (u#NE#v):
          a(v)*xv(u) + b(v)*yv(u) + c(v) = 0
         )
    );
```

Per imporre che i cerchi piccoli siano interamente contenuti nel triangolo bisogna imporre che la loro distanza (con segno) dalla retta del lato sia maggiore o uguale al raggio (che è unitario). La distanza *d* di un punto di coordinate *(xc,yc)* da una retta di parametri *(a,b,c)* è data da:

$$d=\frac{a\,xc+b\,yc+c}{\sqrt{a^2+b^2}}$$

Grazie alla condizione di normalizzazione imposta in precedenza, il denominatore è sempre pari a 1. Quindi i vincoli (uno per ogni coppia retta-cerchio) diventano:

$$a(v)xc(i)+b(v)yc(i)+c(v)\geq 1\,\forall v=1..3\,\forall i=1..K$$

Nella sintassi di Lingo:

```
@for(cerchi(i):
     @for(rette(v):
          a(v)*xc(i) + b(v)*yc(i) + c(v) >= 1
         )
    );
```

Le distanze sono espresse con segno, perché devono risultare positive da una parte della retta e negative dall'altra parte. Per assicurarsi che il segno sia quello giusto, si impone che anche il vertice *(xv(v),yv(v))* opposto ad ogni lato *v* soddisfi la stessa disuguaglianza di cui sopra, cioè che la sua distanza con segno dalla retta *(a(v),b(v),c(v))* sia non-negativa. Il termine noto può essere scelto a piacimento purché sia un valore strettamente positivo e non troppo grande, cioè in modo da non rendere attivo il vincolo. Dato che il triangolo deve contenere cerchi di raggio unitario e deve essere il meno oblungo possibile (per minimizzare il raggio del cerchio che lo racchiude) è ragionevole usare, ad esempio, il valore 1.

$$a(v)x\,v(v)+b(v)y\,v(v)+c(v)\geq 1\,\forall v=1..3$$

Nella sintassi di Lingo:

```
@for(rette(v): a(v)*xv(v) + b(v)*yv(v) + c(v) >= 1);
```

Infine è necessario imporre che i vertici del triangolo siano coperti dal cerchio grande, cioè siano a distanza dall'origine minore o uguale a *r*.

$$xv(v)^2 + yv(v)^2 \leq r^2 \;\forall\, v = 1..3$$

Nella sintassi di Lingo:

```
@for(rette(v): xv(v)^2 + yv(v)^2 <= r2);
```

Il problema è ovviamente non-lineare e non è convesso. Per aiutare il solutore a calcolare una soluzione ammissibile (minimo locale), può essere utile inizializzare i valori delle variabili ad una soluzione ammissibile di partenza.