

Esercizio 5: Macchine

Il problema si può formulare con 4 variabili x che rappresentano il numero di macchine prodotte settimanalmente e 5 variabili y che rappresentano le quantità di componenti elettronici acquistati ogni settimana.

La funzione obiettivo richiede di massimizzare la differenza tra i ricavi settimanali dati dalla vendita delle macchine (combinazione lineare delle variabili x) e i costi settimanali dovuti dall'acquisto dei componenti elettronici (combinazione lineare delle variabili y).

In Lingo la funzione obiettivo è stata formulata nel modo seguente:

```
max 2050 x1 + 3500 x2 + 2550 x3 + 1250 x4 - 30 y1 - 40 y2 - 35 y3 - 10 y4 - 15 y5
```

I vincoli relativi alla produzione impongono che le quantità di componenti utilizzati non siano superiori a quelle acquistate.

Questi sono i vincoli relativi alla produzione:

```
r1) 3 x1 + 4 x2 + 1 x3 + 1 x4 - y1 <= 0
r2) 2 x1 + 2 x2 + 3 x3 + 2 x4 - y2 <= 0
r3) 2 x1 + 5 x2 + 3 x3 + 0 x4 - y3 <= 0
r4) 0 x1 + 4 x2 + 3 x3 + 1 x4 - y4 <= 0
r5) 3 x1 + 4 x2 + 2 x3 + 1 x4 - y5 <= 0
```

Esistono anche vincoli relativi alla minima produzione per ciascuno dei 4 tipi di macchina.

I vincoli relativi alla minima produzione, in Lingo sono stati espressi nel seguente modo:

```
xmin1) x1 >= 25
xmin2) x2 >= 15
xmin3) x3 >= 35
xmin4) x4 >= 30
```

Infine un ulteriore vincolo implica che la somma dei tempi di collaudo non ecceda la quantità di manodopera a disposizione.

Ecco come è stato posto in Lingo il vincolo sui tempi di collaudo:

```
tempo) 10 y1 + 15 y2 + 10 y3 + 20 y4 + 25 y5 <= 21600
```

Si è posto il tutto minore o uguale di 21600 perché questa cifra è il numero di minuti di lavoro disponibili per ogni operaio per ogni settimana.

Il problema è di programmazione lineare nel continuo, con 9 variabili tutte continue e non negative, e 10 vincoli di disuguaglianza il modello è nel file MACCHINE.LTX.

```
! Variabili x(j) = numero di macchine prodotte ogni settimana, j=1..4
! Variabili y(i) = numero componenti comprati ogni settimana, i=1..5
! Tutte le variabili sono continue e non-negative
```

```
! Funzione obiettivo: massimizzazione profitti = ricavi - costi [Euro / settimana]
```

```
max 2050 x1 + 3500 x2 + 2550 x3 + 1250 x4 - 30 y1 - 40 y2 - 35 y3 - 10 y4 - 15 y5
```

```
st
```

```

! Vincoli sulla produzione minima [numero macchine / settimana]

xmin1) x1 >= 25
xmin2) x2 >= 15
xmin3) x3 >= 35
xmin4) x4 >= 30

! Vincoli sull'uso dei componenti [numero componenti / settimana]

r1) 3 x1 + 4 x2 + 1 x3 + 1 x4 - y1 <= 0
r2) 2 x1 + 2 x2 + 3 x3 + 2 x4 - y2 <= 0
r3) 2 x1 + 5 x2 + 3 x3 + 0 x4 - y3 <= 0
r4) 0 x1 + 4 x2 + 3 x3 + 1 x4 - y4 <= 0
r5) 3 x1 + 4 x2 + 2 x3 + 1 x4 - y5 <= 0

! Vincolo sul tempo di lavorazione [minuti / settimana]
! Il termine noto è calcolato come prodotto tra il numero di operai e
! il numero di minuti di lavoro disponibili per ogni operaio per ogni settimana
! 60 [min/h] * 40 [h/sett] * 9 [operai] = 21600 [min/sett]

tempo) 10 y1 + 15 y2 + 10 y3 + 20 y4 + 25 y5 <= 21600

end

```

Dei quattro tipi di macchine i primi tre non sono convenienti: infatti la loro produzione è al minimo necessario (i primi tre vincoli sulla produzione minima sono attivi). Quindi contratti che prevedano una diminuzione delle minime quantità da produrre per quei tre tipi sarebbero vantaggiosi per l'azienda. Non servirebbe invece rendere meno stringente il vincolo sul quarto tipo di macchina, poiché il vincolo non è attivo. Il tutto lo si nota nel file MACCHINE.OUT.

VARIABLE	VALUE	REDUCED COST
X1	25.000000	0.000000
X2	15.000000	0.000000
X3	35.000000	0.000000
X4	75.294121	0.000000
Y1	245.294113	0.000000
Y2	335.588226	0.000000
Y3	230.000000	0.000000
Y4	240.294113	0.000000
Y5	280.294128	0.000000

ROW	SLACK OR SURPLUS	DUAL PRICES
XMIN1)	0.000000	-268.235291
XMIN2)	0.000000	-910.294128
XMIN3)	0.000000	-322.941162
XMIN4)	45.294117	0.000000
R1)	0.000000	161.176468
R2)	0.000000	236.764709
R3)	0.000000	166.176468
R4)	0.000000	272.352936
R5)	0.000000	342.941162
TEMPO)	0.000000	13.117647

Anche senza ricorrere all'analisi parametrica sul vincolo sui tempi di collaudo si può notare dall'analisi di sensitività che il termine noto di tale vincolo (proporzionale al numero di operai) può aumentare indefinitamente senza che la base ottima cambi. Quindi il prezzo-ombra del vincolo resta sempre pari a 13.117647. Quindi poiché ogni operaio contribuisce al termine noto del vincolo per un totale di 2400 minuti/settimana, l'azienda aumenterebbe il profitto di $13.117647 \cdot 2400 = 31482.35$ Euro / settimana per ogni ulteriore operaio assunto.

Dall'analisi di sensitività si vede che il costo del terzo componente (coefficiente della variabile y_3) può aumentare illimitatamente senza che si verifichino cambi di base ottima. Ciò si spiega osservando che il componente 3 è usato solo nella produzione dei tre tipi di macchina non convenienti e quindi la quantità y_3 è già al valore minimo possibile. Anche se il suo prezzo aumentasse non sarebbe comunque possibile acquistarne di meno senza violare i vincoli. Naturalmente, anche se la base non cambia, il valore ottimo diminuisce, fino ad azzerarsi quando l'aumento del costo di y_3 è pari a z^*/y_3^* cioè $251677.9 / 230 = 1094.25$ Euro/pezzo circa.