

Esercizio 3: Forno

Il problema si formula con un modello di programmazione lineare con 4 variabili continue non-negative che rappresentano le quantità prodotte per ciascuno dei 4 tipi di alimentari.

Quattro vincoli definiscono i valori minimi di produzione e un ulteriore vincolo rappresenta le risorse limitate a disposizione, che in questo caso consistono nell'area del forno e nel tempo: poiché il forno funziona per 6 ore ed è ampio 4 mq, le risorse disponibili consistono in 24 mq*h. Tale è il valore del termine noto del vincolo.

In Lingo il vincolo sulle risorse è stato espresso nel seguente modo:

```
ris) 1 x1 + 0.6 x2 + 2.25 x3 + 3 x4 <= 24
```

Il primo membro del vincolo è invece una combinazione lineare delle 4 variabili, dove ciascuno dei coefficienti è dato dal prodotto del tempo unitario e dell'area unitaria, ossia della quantità di risorsa richiesta da ogni kg di prodotto.

Vincoli sulla produzione minima:

```
p_min1) x1 >= 5  
p_min2) x2 >= 2  
p_min3) x3 >= 3  
p_min4) x4 >= 2
```

Il vincolo sulle proporzioni si può esprimere imponendo che esista un valore, m , maggiore o uguale a tutti e 4 i valori delle variabili x e minore o uguale al doppio di ciascuna di esse.

Ecco come sono stati posti i vincoli sulle proporzioni in Lingo:

```
2x1 - m >= 0  
2x2 - m >= 0  
2x3 - m >= 0  
2x4 - m >= 0  
x1 - m <= 0  
x2 - m <= 0  
x3 - m <= 0  
x4 - m <= 0
```

La funzione obiettivo ovviamente, anche se non specificato esplicitamente nel testo, è la massimizzazione dei ricavi che si ottengono dalla vendita dei prodotti.

Funzione obiettivo così come è stata formulata in Lingo:

```
max 2.5 x1 + 4.0 x2 + 2.0 x3 + 5.5 x4
```

Il problema si formula con un modello di programmazione lineare con 4 variabili continue non-negative che rappresentano le quantità prodotte per ciascuno dei 4 tipi di alimentari.

Il modello è nel file FORNO.LTX.

```
! Variabili: x(i)= quantità giornaliera prodotte i=1..4 [Kg]  
!  
! m = massima produzione giornaliera [Kg]
```

```

! Funzione obiettivo: massimizzare i ricavi giornalieri [Euro]
max 2.5 x1 + 4.0 x2 + 2.0 x3 + 5.5 x4

st

! Vincoli sulla produzione minima [Kg]
p_min1) x1 >= 5
p_min2) x2 >= 2
p_min3) x3 >= 3
p_min4) x4 >= 2

! Vincolo di risorsa [h*mq]
ris) 1 x1 + 0.6 x2 + 2.25 x3 + 3 x4 <= 24

! Vincolo sulle proporzioni [Kg]
2x1 - m >= 0
2x2 - m >= 0
2x3 - m >= 0
2x4 - m >= 0
x1 - m <= 0
x2 - m <= 0
x3 - m <= 0
x4 - m <= 0

end

```

Dall'analisi di sensitività sui termini noti, si vede che il prezzo dei prodotti 1, 3 e 4 (pane, focacce e paste) può essere dimezzato senza che cambi la produzione ottima. La soluzione calcolata la Lindo è nel file FORNO.OUT.

VARIABLE	VALUE	REDUCED COST
X1	5.000000	0.000000
X2	5.833333	0.000000
X3	3.000000	0.000000
X4	2.916667	0.000000
M	5.833333	0.000000

ROW	SLACK OR SURPLUS	DUAL PRICES
P_MIN1)	0.000000	-0.714286
P_MIN2)	3.833333	0.000000
P_MIN3)	0.000000	-5.232143
P_MIN4)	0.916667	0.000000
RIS)	0.000000	3.214286
7)	4.166667	0.000000
8)	5.833333	0.000000
9)	0.166667	0.000000
10)	0.000000	-2.071429
11)	0.833333	0.000000
12)	0.000000	2.071429
13)	2.833333	0.000000
14)	2.916667	0.000000

L'analisi parametrica sul vincolo relativo alle risorse di tempo e area rivela il prezzo ombra del vincolo. Poiché l'area del forno misurata in mq è pari a 4, un incremento unitario del termine noto del vincolo corrisponde a $\frac{1}{4}$ d'ora di lavoro straordinario e quindi a un costo di 3 Euro. Inizialmente il prezzo-ombra del vincolo è superiore a 3 Euro (è pari a 3.21 Euro circa), quindi conviene il lavoro straordinario. Dopo il primo cambio di base però ($\text{rhs}=24.35$) il prezzo-ombra scende al valore di 2.5 Euro e quindi ulteriore lavoro straordinario non conviene più. Quindi la quantità di lavoro straordinario più conveniente è pari a circa 0.09 ore, cioè circa 5.4 minuti.